********************************************************************************
** Title: Replication for analysis in "Public trust, policing, and the COVID-19 pandemic." 2022. SSM 305(115045). 
** Authors: Robert Blair, Travis Curtice, David Dow, Guy Grossman
** Version: July 14, 2022 
********************************************************************************

** Set directory
cd "C:\Users\ddow1\Dropbox\Covid-Uganda\SSM Replication"

* Load dataset
use "ug_covid_ssm_replication.dta", clear

set more off
set scheme plottig

********************************************************************************
****** Survey Endorsement Experiments 
********************************************************************************

** Randomize endorsing institution
egen zf_agree_gather = std(f_agree_gather)
egen zf_agree_distance = std(f_agree_distance)

** construct policy mean agreement index for gather and distance only
gen covid_policy_avgindex = (f_agree_gather + f_agree_distance)/2
egen zcovid_policy_avgindex = std(covid_policy_avgindex)

gen sample_endorse = 1 if zcovid_policy_avgindex != .

********************************************************************************
** Figure 1 - Endorsement Experiment: Plot Predicted Values 
********************************************************************************

quietly reg f_agree_gather i.exp_instn if sample_endorse==1, cluster(village_id_num)
margins exp_instn
marginsplot, title(Support for Banning Social Gatherings) recast(scatter) xscale(range(.75 4)) xlabel(, angle(45)) xtitle("")
graph save gather.gph, replace

quietly reg f_agree_distance i.exp_instn if sample_endorse==1, cluster(village_id_num)
margins exp_instn
marginsplot, title(Support for Social Distancing) recast(scatter) xscale(range(.75 4)) xlabel(, angle(45)) xtitle(Endorsing Institution)
graph save distance.gph, replace

quietly reg covid_policy_avgindex i.exp_instn if sample_endorse==1, cluster(village_id_num)
margins exp_instn
marginsplot, title(Support for Policy Index) recast(scatter) xscale(range(.75 4)) xlabel(, angle(45)) xtitle("")
graph save index.gph, replace

graph combine gather.gph distance.gph index.gph, ycommon xcommon rows(1) title(Figure 1: Endorsement Experiment)
graph save figure1.gph, replace
graph export figure1.pdf, replace

********************************************************************************
* Table D.1 - Endorsement Experiments OLS Regressions
********************************************************************************
eststo clear
eststo: quietly reg zf_agree_gather i.exp_instn if sample_endorse==1, cluster(village_id_num)
eststo: quietly reg zf_agree_distance i.exp_instn if sample_endorse==1, cluster(village_id_num)
eststo: quietly reg zcovid_policy_avgindex i.exp_instn if sample_endorse==1, cluster(village_id_num)

esttab using table_d1_surv_exp_inst.tex, label nogap noomitted b(2) se(2) mtitle("Support Ban Gatherings" "Support Social Distancing" "Index of Support") title(Table D.1 Survey Experiment - Endorsing Institution\label{tab:institutions}) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

********************************************************************************
** Figure D.1: Support for Banning Gatherings (Pairwise Comparisons)
********************************************************************************
reg zf_agree_gather i.exp_instn if sample_endorse==1, cluster(village_id_num)
margins exp_instn, pwcompare mcompare(noadjust)
marginsplot, horizontal unique xline(0) recast(scatter) title("Figure D.1: Support for Banning Gatherings (Pairwise Comparisons)") yscale(reverse)  ylabel(1 "INGO vs Govt" 2 "Religious vs Govt" 3 "Traditional vs. Govt" 4 "Religion vs INGO" 5 "Traditional vs. INGO" 6 "Traditional vs INGO")
graph save fig_d1_endorsement_pairwise_nc.gph, replace
graph export fig_d1_endorsement_pairwise_nc.pdf, replace


********************************************************************************
** Table D.2: Endorsement Experiment - Balance Tests
********************************************************************************

label var bcpincome_hh2 "Income"
label var bcptrust_community "Trust Community"

iebaltab gender_all age_all education_all b_before_religion nrm_id bcpincome_hh2 bcplegit_trust bcpsatis_trust bcplegit_lc5_trust bcplegit_lc3_trust bcptrust_lc1 bcptrust_community, grpvar(exp_instn) feqtest nottest rowvarlabels pftest tblnonote savetex("table_d2_endorsement_balance.tex") replace

********************************************************************************
**** Observational Analyses 
********************************************************************************

** create summary compliance index 
icw_index zhome zpray zcafe zmeeting zsocial zdist zmask, gen(compliance_icw)
egen compliance_icw_st = std(compliance_icw)

** Standardize Trust Measures
local vars_st bcplegit_lc5_trust bcplegit_trust bcpsatis_trust bcptrust_community bcptrust_lc1 bcplegit_lc3_trust d_trust_government d_trust_moh d_trust_police d_trust_district d_trust_lc1 d_trust_people bcpage  bcpincome_hh2 bcpeducation
foreach v in `vars_st' {
	quietly egen z`v' = std(`v')
}

* Label baseline measures
label var zbcplegit_trust "Trust Central Govt"
label var zbcpsatis_trust "Trust Police"
label var zbcplegit_lc5_trust "Trust District Govt"
label var zbcptrust_lc1 "Trust LC1 Govt"
label var zbcptrust_community "Trust People in Community"
label var zbcplegit_lc3_trust "Trust in LC3 Govt"

** Label follow-ups 
label var zd_trust_government "Trust Central Govt"
label var zd_trust_moh "Trust MoH"
label var zd_trust_police "Trust Police"
label var zd_trust_district "Trust District Govt"
label var zd_trust_lc1 "Trust LC1"
label var zd_trust_people "Trust People in Community"

** Label controls 
label var bcpmale "Female"
label var zbcpage "Age"
label var zbcpeducation "Education"
label var zbcpincome_hh2 "Income (Standardized)"

** set control variables
local bdemos bcpmale zbcpage zbcpeducation zbcpincome_hh2

** Create Standardized Trust in Government Index
** Baseline trust
alpha bcplegit_lc3_trust bcplegit_lc5_trust bcplegit_trust bcpsatis_trust, std item c generate(bcp4_gov_trust_index)
egen zbcp4_gov_trust_index = std(bcp4_gov_trust_index)

** Follow-up trust
alpha d_trust_district d_trust_government d_trust_police d_trust_moh, std item c generate(fu_gov_trust_index)
egen zfu_gov_trust_index = std(fu_gov_trust_index)

label var bcp4_gov_trust_index "Trust in Govt (Index)"
label var zbcp4_gov_trust_index "Trust in Govt (Index)"
label var fu_gov_trust_index "Trust in Govt (Index)"
label var zfu_gov_trust_index "Trust in Govt (Index)"

** Set baseline and follow-up trust variables to use in regressions
local trust_index_bcp zbcp4_gov_trust_index zbcptrust_lc1 zbcptrust_community
local trust_index_fu zfu_gov_trust_index zd_trust_lc1 zd_trust_people

********************************************************************************
****  Main analysis for trust in government index
********************************************************************************

** Table 1: Correlates of Public Health Compliance
** Top Panel: Baseline measures of trust (trust in state index, lc1, and trust in community)

eststo clear
foreach v in `trust_index_bcp' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_index_bcp' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table1_compliance_baseline_index_trust.tex, label nogap b(2) se(2) mtitle("" "" "" "") title(Table 1: Correlates of Public Health Compliance - Baseline\label{tab:bcptrust}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

** Bottom Panel: Follow-up measures of trust (trust in state index, lc1, and trust in community)
eststo clear 
foreach v in `trust_index_fu' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_index_fu'  `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table1_compliance_followup_index_trust.tex, label nogap b(2) se(2) mtitle("" "" "" "") title(Table 1: Correlates of Public Health Compliance - Mobile Survey\label{tab:ftrust}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


** Test for differences between coefficients
	** Baseline measures

eststo clear
foreach v in `trust_index_bcp' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1
}

suest est1 est2, cluster(village_id_num)
test [est1_mean]zbcp4_gov_trust_index-[est2_mean]zbcptrust_lc1 = 0

suest est1 est3, cluster(village_id_num)
test [est1_mean]zbcp4_gov_trust_index-[est3_mean]zbcptrust_community = 0

	** Follow-up measures
eststo clear 
foreach v in `trust_index_fu' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1
}

suest est1 est2, cluster(village_id_num)
test [est1_mean]zfu_gov_trust_index-[est2_mean]zd_trust_lc1 = 0

suest est1 est3, cluster(village_id_num)
test [est1_mean]zfu_gov_trust_index-[est3_mean]zd_trust_people = 0



********************************************************************************
** Table D.4: Correlates of Public Health Compliance (with Partisanship)
********************************************************************************

** Top panel - baseline trust measures with partisanship
eststo clear
foreach v in `trust_index_bcp' {
	eststo: quietly reg compliance_icw_st `v' nrm_id `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_index_bcp' nrm_id `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d4_baseline.tex, label nogap b(2) se(2) mtitle("" "" "" "") title(Table D4: Correlates of Public Health Compliance (with Partisanship)\label{tab:bcptrustnrm}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

** Bottom panel - follow-up trust measures with partisanship
eststo clear 
foreach v in `trust_index_fu' {
	eststo: quietly reg compliance_icw_st `v' nrm_id `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_index_fu' nrm_id `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d4_followup.tex, label nogap b(2) se(2) mtitle("" "" "" "") title(Table D4: Correlates of Public Health Compliance - Follow-Up Survey\label{tab:ftrustnrm}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


********************************************************************************
** Table 2: Disaggregating the trust in state index individually
********************************************************************************
* set baseline and follow-up measures 
local trust_vars_bcp zbcplegit_trust zbcpsatis_trust zbcplegit_lc5_trust zbcplegit_lc3_trust
local trust_vars_fu zd_trust_government zd_trust_police zd_trust_district zd_trust_moh

** Top panel - baseline measures disaggregated
eststo clear 
foreach v in `trust_vars_bcp' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_vars_bcp' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table2_baseline.tex, label nogap b(2) se(2) mtitle("" "" "" "" "") title(Table 2: Disaggregating Trust in the State - Baseline\label{tab:bcptrust}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


** Bottom panel - Follow-up measures disaggregated
eststo clear 
foreach v in `trust_vars_fu' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_vars_fu' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table2_followup.tex, label nogap b(2) se(2) mtitle("" "" "" "" "" "") title(Table 2: Disaggregating Trust in the State - Follow-Up\label{tab:bcptrust}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


********************************************************************************
** Table D.5: Correlates of Public Health Compliance (with Partisanship)
********************************************************************************
	** Baseline disaggregate
eststo clear 
foreach v in `trust_vars_bcp' {
	eststo: quietly reg compliance_icw_st `v' nrm_id `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_vars_bcp' nrm_id `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d5_baseline.tex, label nogap b(2) se(2) mtitle("" "" "" "" "") title(Table D.5: Disaggregating Trust in the State - Baseline (With NRM ID)\label{tab:bcpdistrustnrm}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

	** Follow-up disaggregate

eststo clear 
foreach v in `trust_vars_fu' {
	eststo: quietly reg compliance_icw_st `v' nrm_id `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_vars_fu' nrm_id `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d5_followup.tex, label nogap b(2) se(2) mtitle("" "" "" "" "" "") title(Table D.5: Disaggregating Trust in the State - Follow-Up with NRM ID\label{tab:distrustnrm}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


********************************************************************************
** Table D.6: Correlates of Public Health Compliance (No Controls)
********************************************************************************

** Top panel: BASELINE Trust index no controls
eststo clear 
foreach v in `trust_index_bcp' {
	eststo: quietly reg compliance_icw_st `v' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_index_bcp' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d6_baseline.tex, label nogap b(2) se(2) mtitle("" "" "" "") title(Table D6: Correlates of Public Health Compliance - Baseline (No Controls)\label{tab:bcptrust}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

** Bottom panel: FOLLOWUP index - no controls 
eststo clear 
foreach v in `trust_index_fu' {
	eststo: quietly reg compliance_icw_st `v' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_index_fu' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d6_followup.tex, label nogap b(2) se(2) mtitle("" "" "" "") title(Table D6: Correlates of Public Health Compliance - Follow-Up (No Controls)\label{tab:bcptrust}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


********************************************************************************
** Table D.7: Disaggregating Trust in the State (No Controls)
********************************************************************************
** Top panel: BASELINE Trust disaggregated no controls
eststo clear 
foreach v in `trust_vars_bcp' {
	eststo: quietly reg compliance_icw_st `v' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_vars_bcp' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d7_baseline.tex, label nogap b(2) se(2) mtitle("" "" "" "" "" "") title(Table D7: Disaggregating Trust in the State - Baseline (No Controls)\label{tab:dis_controls_bcp}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

** Bottom panel: Follow-up Trust disaggregated no controls
eststo clear 
foreach v in `trust_vars_fu' {
	eststo: quietly reg compliance_icw_st `v' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg compliance_icw_st `trust_vars_fu' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d7_followup.tex, label nogap b(2) se(2) mtitle("" "" "" "" "" "") title(Table D7: Disaggregating Trust in the State - Follow-Up (No Controls)\label{tab:dis_controls_fu}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


********************************************************************************
** Table D.8: Correlates of Public Health Compliance - District Clustered S.E.
********************************************************************************
** Top panel: BASELINE Trust
eststo clear
foreach v in `trust_index_bcp' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(district)
}
eststo: quietly reg compliance_icw_st `trust_index_bcp' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(district)
esttab using table_d8_baseline.tex, label nogap b(2) se(2) mtitle("" "" "" "") title(Table D8: Correlates of Public Health Compliance - Baseline\label{tab:observational_dist}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

** Bottom panel: Follow-up Trust
eststo clear 
foreach v in `trust_index_fu' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(district)
}
eststo: quietly reg compliance_icw_st `trust_index_fu'  `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(district)
esttab using table_d8_followup.tex, label nogap b(2) se(2) mtitle("" "" "" "") title(Table D8: Correlates of Public Health Compliance - Follow-Up Survey\label{tab:observational_dist}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


********************************************************************************
** Table D.9: Disaggregating Trust in the State - District Clustered S.E.
********************************************************************************

** Top panel: BASELINE Trust
eststo clear 
foreach v in `trust_vars_bcp' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(district)
}
eststo: quietly reg compliance_icw_st `trust_vars_bcp' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(district)
esttab using table_d9_baseline.tex, label nogap b(2) se(2) mtitle("" "" "" "" "") title(Table D9: Disaggregating Trust in the State - District Clustered S.E. Baseline\label{tab:disag_dist}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

** Bottom panel: Follow-up Trust
eststo clear 
foreach v in `trust_vars_fu' {
	eststo: quietly reg compliance_icw_st `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(district)
}
eststo: quietly reg compliance_icw_st `trust_vars_fu' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(district)
esttab using table_d9_followup.tex, label nogap b(2) se(2) mtitle("" "" "" "" "" "") title(Table D9: Disaggregating Trust in the State - District Clustered S.E. Follow-Up\label{tab:disag_dist}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

*********************************************************************************
** Table D.10: Correlates of COVID Knowledge (True/False Questions)
*********************************************************************************

** Baseline

eststo clear
foreach v in `trust_index_bcp' {
	eststo: quietly reg tf_correct_sum `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg tf_correct_sum `trust_index_bcp' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d10_baseline.tex, label nogap noomitted nobaselevels b(2) se(2) mtitle("" "" "" "") title(Table D10: Correlates of COVID Knowledge (True/False Questions)\label{tab:tfbase}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

** FOLLOWUP
eststo clear 
foreach v in `trust_index_fu' {
	eststo: quietly reg tf_correct_sum `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg tf_correct_sum `trust_index_fu' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d10_followup.tex, label nogap noomitted nobaselevels b(2) se(2) mtitle("" "" "" "") title(Table D10: Correlates of COVID Knowledge (True/False Questions)\label{tab:tffollow}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


*********************************************************************************
** Table D.11: Correlates of Knowledge of COVID Symptoms
*********************************************************************************
** Baseline trust
eststo clear
foreach v in `trust_index_bcp' {
	eststo: quietly reg sum_symptoms `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg sum_symptoms `trust_index_bcp' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d11_baseline.tex, label nogap noomitted nobaselevels b(2) se(2) mtitle("" "" "" "") title(Table D11: Correlates of Knowledge of COVID Symptoms - Baseline\label{tab:symptoms}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

** FOLLOWUP trust
eststo clear 
foreach v in `trust_index_fu' {
	eststo: quietly reg sum_symptoms `v' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
}
eststo: quietly reg sum_symptoms `trust_index_fu' `bdemos' village_id_dummy* if base_follow_sample == 1, cluster(village_id_num)
esttab using table_d11_followup.tex, label nogap noomitted nobaselevels b(2) se(2) mtitle("" "" "" "") title(Table D11: Correlates of Knowledge of COVID Symptoms- Follow-Up\label{tab:symptoms}) drop(village_id_dummy*) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace


*********************************************************************************
** Table D.3 Correlations in Measures of Trust
*********************************************************************************
** Top panel: baseline measures of trust
eststo clear
estpost correlate zbcplegit_trust zbcpsatis_trust zbcplegit_lc5_trust zbcplegit_lc3_trust zbcptrust_lc1 zbcptrust_community nrm_id if base_follow_sample == 1, matrix listwise
est store m1
esttab m1 using table_d3_baseline.tex, title(Table D.3: Correlation Matrix - Baseline Survey) unstack nostar not noobs label compress replace

** Bottom panel: follow-up measures of trust
estpost correlate zd_trust_government zd_trust_police zd_trust_district zd_trust_moh zd_trust_lc1 zd_trust_people nrm_id if base_follow_sample == 1, matrix listwise
est store m2
esttab m2 using table_d3_followup.tex, title(Table D.3: Correlation Matrix - Follow-Up Survey) unstack nostar not noobs label compress replace

*********************************************************************************
** Table A.1 Summary statistics
*********************************************************************************

* Rescale raw baseline trust measures to match follow-up
replace bcplegit_trust = bcplegit_trust + 1
replace bcplegit_lc5_trust = bcplegit_lc5_trust + 1
replace bcplegit_lc3_trust =  bcplegit_lc3_trust + 1
replace bcptrust_lc1 = bcptrust_lc1 + 1
replace bcptrust_community = bcptrust_community + 1

label var bcplegit_trust "Trust Central Gov (Baseline)"
label var bcpsatis_trust "Trust Police (Baseline)"
label var bcplegit_lc5_trust "Trust District Gov (Baseline)"
label var bcptrust_lc1 "Trust LC1 Gov (Baseline)"
label var bcptrust_community "Trust People in Community (Baseline)"
label var bcplegit_lc3_trust "Trust in LC3 Gov (Baseline)"
label var d_trust_government "Trust Central Gov (Follow-up)"
label var d_trust_moh "Trust MoH (Follow-up)"
label var d_trust_police "Trust Police (Follow-up)"
label var d_trust_district "Trust District Gov (Follow-up)"
label var d_trust_lc1 "Trust LC1 Gov (Follow-up)"
label var d_trust_people "Trust People in Community (Follow-up)"

eststo clear
estpost summarize b_times_home b_times_pray b_times_cafe b_times_meeting b_times_social b_keep_distance b_face_mask f_agree_lockdown f_agree_gather f_agree_distance bcplegit_trust bcpsatis_trust bcplegit_lc5_trust bcplegit_lc3_trust bcptrust_lc1 bcptrust_community d_trust_government d_trust_moh d_trust_police d_trust_district d_trust_lc1 d_trust_people nrm_id sum_symptoms tf_correct_sum gender_all age_all education_all b_before_religion

esttab using table_a1_sumstats_scaled.tex, label nogap cells("count(fmt(0)) mean(fmt(2)) sd(fmt(2)) min(fmt(1)) max(fmt(1))") title(Table A1: Summary Statistics\label{tab:sumstats}) replace

********************************************************************************
** List Experiments
********************************************************************************
net install st0559.pkg

** Table D.12 List experiment on not wearing masks
eststo clear
eststo: quietly kict ls list_mask_count, nnonkey(3) condition(list_mask_cond) estimator(linear) vce(cluster village_id_num)
esttab using table_d12.tex, label nogap noomitted nobaselevels b(2) se(2) mtitle("" "" "") title(Table D.12: Text Messaging Experiment - List Experiment on Not Wearing Masks\label{tab:listmask}) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace

** Table D.13 for attending religious service
eststo clear
eststo: quietly kict ls list_relig_count, nnonkey(3) condition(list_relig_cond) estimator(linear) vce(cluster village_id_num)
esttab using table_d13.tex, label nogap noomitted nobaselevels b(2) se(2) mtitle("" "" "") title(Table D.13: Text Messaging Experiment - List Experiment on Attending Religious Service\label{tab:listrelig}) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) replace